
	Reason for change:
	The soundingRS-UL-Config dedicated determines uplink SRS configuration for UE in RRC connected mode. Before the UE establishes RRC connection it shall apply default physical channel configuration as specified in 36.331, cl. 9.2.4 which dictates the SoundingRS-UL-ConfigDedicated to be “release”. 
Additionally, the 36.508, table 4.8.2.1.6 specifies for the soundingRS-UL-ConfigDedicated to be “Not present” in case of “configuration of SRB1 during RRC connection (re-)establishment”. 

However the current TTCN implementation configures the dedicated uplink SRS configuration in the SS at the initialisation of every test and keeps it that way regardless of RRC connection state during the entire test. 
In order to have same physical channel configuration for both UE and SS across both RRC connection states, we propose to configure “release” in the soundingRS-UL-Config dedicated during RRC Idle (i.e. also at the initialisation of the test) and have it set-up for RRC Connected state only.

	
	

	Summary of change:
	The soundingRS-UL-Config dedicated is configured as “release” during RRC Idle and set up during RRC Connected state.

	
	

	Consequences if not approved:
	Physical channel configuration will not be same for UE and SS during RRC Idle state.

	
	

	Clauses affected:
	

	
	

	
	Y
	N
	
	

	Other specs
	
	X
	 Other core specifications

	TS/TR ... CR ... 

	affected:
	
	X
	 Test specifications
	TS/TR ... CR ... 

	(show related CRs)
	
	X
	 O&M Specifications
	TS/TR ... CR ... 

	
	

	Other comments:
	


Change 1
	New template
	cs_508_SoundingRsUl_ConfigDedicated_Release

	Reason for change
	New template to release SoundingRS uplink config dedicated needs to be created for use in RRC Idle state.

	Summary of change
	New template to release SoundingRS uplink config dedicated has been created.

	Source of change
	Common\EUTRA_Templates\EUTRA_RRC_Templates.ttcn


After change :

	  template (value) SoundingRS_UL_ConfigDedicated cs_508_SoundingRsUl_ConfigDedicated_Release :=

  { /* 36.508 Table 4.6.3-22: SoundingRsUl-ConfigDedicated-DEFAULT */

    /* @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS) */

    release := NULL

  };



Change 2

	Function name
	f_EUTRA_SS_ConfigureActiveCell

	Reason for change
	During the initial configuration of SS, in accordance with the proposed change in this CR, the SoundingRS uplink config dedicated needs to be configured as “released”, so it will have the same configuration as UE prior beginning of the test. 

	Summary of change
	During the initial configuration of SS the SoundingRS uplink config dedicated has been configured as “released”.

	Source of change
	Common\EUTRA\EUTRA_ConfigurationSteps.ttcn


Before change:

	  function f_EUTRA_SS_ConfigureActiveCell(EUTRA_CellId_Type p_CellId,

                                          template (omit) CA_CellInitialConfig_Type p_CA_CellInitialConfig := omit) runs on EUTRA_PTC

  {

    var EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get (p_CellId);

    var template (value) AllSiSchedul_Type    v_SiSchedul;

    var template (value) ContentionResolution_ContainedRlcPdu_Type v_ContentionResolution_ContainedRlcPdu;

    var template (value) DL_CCCH_Message v_RrcConnSetup;

    var octetstring v_EncodedRrcConnSetup;

    //Get FDD or TDD mode from Cell-configuration

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (p_CellId);

    var template (value) StaticCellInfo_Type v_StaticCellInfo;

    var template (omit) TDD_Config_v1130 v_TDD_Config_v1130 := omit;

    v_SiSchedul := f_EUTRA_SS_SetupSchedulingInfo(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1, v_EUTRA_FDD_TDD_Mode);      /* derive SS's scheduling information from SIB1 */
…




After change :

	 function f_EUTRA_SS_ConfigureActiveCell(EUTRA_CellId_Type p_CellId,

                                          template (omit) CA_CellInitialConfig_Type p_CA_CellInitialConfig := omit) runs on EUTRA_PTC

  {

    var EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get (p_CellId);

    var template (value) AllSiSchedul_Type    v_SiSchedul;

    var template (value) ContentionResolution_ContainedRlcPdu_Type v_ContentionResolution_ContainedRlcPdu;

    var template (value) DL_CCCH_Message v_RrcConnSetup;

    var octetstring v_EncodedRrcConnSetup;

    //Get FDD or TDD mode from Cell-configuration

    var EUTRA_FDD_TDD_Mode_Type  v_EUTRA_FDD_TDD_Mode := f_EUTRA_CellInfo_GetFDD_TDD (p_CellId);

    var template (value) StaticCellInfo_Type v_StaticCellInfo;

    var template (omit) TDD_Config_v1130 v_TDD_Config_v1130 := omit;

    v_CellInfo.FDD_TDD_Info.soundingRS_UL_Config := cs_508_SoundingRsUl_ConfigDedicated_Release;
    v_SiSchedul := f_EUTRA_SS_SetupSchedulingInfo(v_CellInfo.Sysinfo.BCCH_Info.SIB1.message_.c1.systemInformationBlockType1, v_EUTRA_FDD_TDD_Mode);      /* derive SS's scheduling information from SIB1 */
...




Change 3

	New template
	cas_CellConfig_SRS

	Reason for change
	New template to configure SS with just SoundingRS uplink config dedicated needs to be created so the configuration can be changed during transition between RRC states.

	Summary of change
	New template to configure SS with SoundingRS uplink config dedicated has been created.

	Source of change
	Common\EUTRA_Templates\EUTRA_CellCfg_Templates.ttcn


After change :

	  template (value) SYSTEM_CTRL_REQ cas_CellConfig_SRS(EUTRA_CellId_Type  p_CellId,

                                                               template (value) TimingInfo_Type p_TimingInfo,

                                                               EUTRA_CellInfo_Type p_CellInfo

                                                               ) :=

  { /* @status    APPROVED (LTE_A) */

    Common := cs_ReqAspCommonPart_CellCfg(p_CellId, p_TimingInfo),

    Request := {

      Cell := {

        AddOrReconfigure := {

          Basic := omit,

          Active := {

            C_RNTI := omit,

            PhysicalLayerConfigUL := {

              Prach := omit,

              Pucch := omit,

              Pusch := omit,

              TimingAdvance := omit,

              SRS_UL_Config := {

                Common := cs_SoundingRS_UL_ConfigCommon_r8( p_CellInfo.Sysinfo.BCCH_Info.SIs[0].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib2.radioResourceConfigCommon.soundingRS_UL_ConfigCommon),    

                Dedicated := cs_SoundingRS_UL_ConfigDedicated_r8( p_CellInfo.FDD_TDD_Info.soundingRS_UL_Config)

              },

              SR_Config := omit,

              CQI_ReportConfig := omit,

              UplinkPowerControlCommon := omit,

              UplinkPowerControlDedicated := omit

            },

            RachProcedureConfig := omit,

            CcchDcchDtchConfig := omit,

            ServingCellConfig := omit

          }

        }

      }

    }

  }; 


Change 4

	Function name
	f_EUTRA_RRC_ConnectionSetup_Def

	Reason for change
	During the transition from RRC Idle to RRC connected state the SS needs to be configured with SoundingRS uplink config dedicated.

	Summary of change
	During the transition from RRC Idle to RRC connected state the SS has been configured with SoundingRS uplink config dedicated.

	Source of change
	Common\EUTRA\EUTRA_RRCSteps.ttcn


Before change:

	  function f_EUTRA_RRC_ConnectionSetup_Def(EUTRA_CellId_Type p_CellId,

                                           template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                           RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def) runs on EUTRA_PTC

  {

    if (f_EUTRA_DlCcchMsgInSeparateMacPdu()) {    /* @sic R5-120721 change 3 sic@ */

      // switch on UL grant assignments

      f_EUTRA_StartDefULGrantTransmission(p_CellId);

      SRB.send(cas_SRB0_RrcPdu_REQ(p_CellId,

                                   p_TimingInfo,

                                   f_EUTRA_508_RRCConnectionSetup(p_CellId, p_RRC_TI)));  /* @sic R5-120721 change 3 sic@ */

    }

  }


After change :

	  function f_EUTRA_RRC_ConnectionSetup_Def(EUTRA_CellId_Type p_CellId,

                                           template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                           RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def) runs on EUTRA_PTC

  {

  f_EUTRA_SS_CommonCellConfig(p_CellId, cas_CellConfig_SRS(p_CellId,

                                                           cs_TimingInfo_Now,

                                                           f_EUTRA_CellInfo_Get(p_CellId)));
    if (f_EUTRA_DlCcchMsgInSeparateMacPdu()) {    /* @sic R5-120721 change 3 sic@ */

      // switch on UL grant assignments

      f_EUTRA_StartDefULGrantTransmission(p_CellId);

      SRB.send(cas_SRB0_RrcPdu_REQ(p_CellId,

                                   p_TimingInfo,

                                   f_EUTRA_508_RRCConnectionSetup(p_CellId, p_RRC_TI)));  /* @sic R5-120721 change 3 sic@ */

    }

  }


Change 5

	Function name
	f_EUTRA_RRC_ConnectionRelease_Local

	Reason for change
	During the transition from RRC connected to RRC idle state the SS needs to be configured with the release in SoundingRS uplink config dedicated in order to use the same configuration as UE.

	Summary of change
	During the transition from RRC connected to RRC idle state the SS has been configured with release in SoundingRS uplink config dedicated.

	Source of change
	Common\EUTRA\EUTRA_RRCSteps.ttcn


Before change:

	  function f_EUTRA_RRC_ConnectionRelease_Local(EUTRA_CellId_Type p_CellId,

                                               SubFrameTiming_Type p_TimingAtT,

                                               template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def) runs on EUTRA_PTC

  { /* NOTE: the function schedules all steps to release and reconfigure the SRBs and (default) DRB

     *       => this will in fact happen in the future */

    /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    /* @sic R5s110743: release MeasGapConfig sic@ */

    var SubFrameTiming_Type v_Timing := p_TimingAtT;

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    /* Step 1 at T: disable the UL grant

     *               => it is not possible for the UE to send any ACK for the RRC Connecion Release)

     *               => UE has to wait until 60ms timer has timed out (-> 36.331 cl. 5.3.8.3) */

    //f_EUTRA_StopULGrantTransmission(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number)); //Anritsu TTCN WA RRC release

    /* Step 2 at T+5ms: release security */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 5);

    f_EUTRA_SS_RRC_ReleaseSecurity(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3 at T+10ms: release DRX */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 10);

    f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3A at T+15ms: release MeasGapConfig */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 15);

    f_EUTRA_SS_MeasGapCtrlConfig(p_CellId, cs_MeasGapCtrl_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 4 (obsolete) */

    /* Step 5 at T+55ms: Release SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 55);

    f_EUTRA_SS_SRBs_DRBs_Release(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), v_DRB_IdentityList.All);

    /* Step 6 at T+60ms: Configure SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 60);

    f_EUTRA_SS_SRBs_DRBs_Config(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), p_DrbConfigList);

  }


After change :

	  function f_EUTRA_RRC_ConnectionRelease_Local(EUTRA_CellId_Type p_CellId,

                                               SubFrameTiming_Type p_TimingAtT,

                                               template (value) RadioBearerList_Type p_DrbConfigList := cs_DrbConfigList_Def) runs on EUTRA_PTC

  { /* NOTE: the function schedules all steps to release and reconfigure the SRBs and (default) DRB

     *       => this will in fact happen in the future */

    /* @sic R5-110803 change 13 sic@ */

    /* @sic R5s110634 MCC160 implementation of change 1.1: new parameter p_DrbConfigList sic@ */

    /* @sic R5s110743: release MeasGapConfig sic@ */

    var SubFrameTiming_Type v_Timing := p_TimingAtT;

    var DRB_IdentityList_AllAndAMonly_Type v_DRB_IdentityList;

    var EUTRA_CellInfo_Type v_CellInfo := f_EUTRA_CellInfo_Get (p_CellId);
    // Build DRB identity list from inputed radiobearer config list

    v_DRB_IdentityList := f_EUTRA_DRB_IdentityList(p_DrbConfigList);

    /* Step 1 at T: disable the UL grant

     *               => it is not possible for the UE to send any ACK for the RRC Connecion Release)

     *               => UE has to wait until 60ms timer has timed out (-> 36.331 cl. 5.3.8.3) */

    //f_EUTRA_StopULGrantTransmission(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number)); //Anritsu TTCN WA RRC release

    /* Step 2 at T+5ms: release security */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 5);

    f_EUTRA_SS_RRC_ReleaseSecurity(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3 at T+10ms: release DRX */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 10);

    f_EUTRA_SS_DrxCtrlConfig(p_CellId, cs_DrxCtrl_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 3A at T+15ms: release MeasGapConfig */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 15);

    f_EUTRA_SS_MeasGapCtrlConfig(p_CellId, cs_MeasGapCtrl_None, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number));

    /* Step 4 (obsolete) */    

    /* Step 5 at T+55ms: Release SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 55);

    f_EUTRA_SS_SRBs_DRBs_Release(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), v_DRB_IdentityList.All);

    //Release SRS UL config dedicated

    v_CellInfo.FDD_TDD_Info.soundingRS_UL_Config := cs_508_SoundingRsUl_ConfigDedicated_Release;

    f_EUTRA_SS_CommonCellConfig(p_CellId, cas_CellConfig_SRS(p_CellId,

                                                             cs_TimingInfo_Now,

                                                             v_CellInfo));
    /* Step 6 at T+60ms: Configure SRBs and DRBs */

    v_Timing := f_EUTRA_TimingInfoAdd(p_TimingAtT, 60);

    f_EUTRA_SS_SRBs_DRBs_Config(p_CellId, cs_TimingInfo(v_Timing.SFN.Number, v_Timing.Subframe.Number), p_DrbConfigList);

  }

