
	Reason for change:
	Postamble is modified to allow UE to perform encapsulated Power Down Registration in 1xRTT before perfoming Detach procedure when the UE supports being switched off (“pc_SwitchOnOff” is TRUE).

	
	

	Summary of change:
	New Postamble function is added in order to handle Power Down Registration when “pc_SwitchOnOff” is TRUE.

Change 1
	Function name
	f_TC_13_4_4_1_EUTRA

	Reason for change
	Postamble is modified to allow UE to perform encapsulated Power Down Registration in 1xRTT before perfoming Detach procedure when the UE supports being switched off (“pc_SwitchOnOff” is TRUE).
A similar change submitted for 13.4.4.2 (R5s140804) was accepted in principle but different implementation was proposed by MCC160. This change follows that proposal by creating new functions (changes 2 and 3 below).


	Summary of change
	Postamble function is replaced by a new function that will also handle Power Down Registration when “pc_SwitchOnOff” is TRUE.

	Source of change
	LTE_IRAT\13\MultiLayer_Procedures_EC2K.ttcn


Before change:

	  function f_TC_13_4_4_1_EUTRA ()  runs on EUTRA_PTC

  { /* Pre-registration at 1xRTT and Cell reselection / 1x Zone Registration */

    var IRAT_CoOrd_SysInfo_Type v_CDMA2000SysInfo1;

    var IRAT_CoOrd_SysInfo_Type v_CDMA2000SysInfo2;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell2, -79)

    };

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init( c6  );

    //Set maximum cell powel level for Cell 1

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, -60 );

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell2, -60 );

    // Wait for CDMA2000Sysinfo to fetch color code

    v_CDMA2000SysInfo1 := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );

    v_CDMA2000SysInfo2 := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );

    f_InitialiseSib8_1XRTT_PreRegistration (eutra_Cell1, v_CDMA2000SysInfo1, px_1XRTT_Zone_Timer);

    f_InitialiseSib8_1XRTT_PreRegistration (eutra_Cell2, v_CDMA2000SysInfo2, px_1XRTT_Zone_Timer);

    //Create and configure Cell 1

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    f_EUTRA_CellConfig_Def ( eutra_Cell2 );

    f_EUTRA_SetCellPower( eutra_Cell1, -85 );

    //Preamble to ensure USIM is updated with valid TAI and to bring UE to initial state

    f_EUTRA_Preamble_CDMA2000_PreRegistration(eutra_Cell1, STATE2_IDLEUPDATE, true, px_RAND);

    //Send trigger to inform start of Test Body

    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );

    f_EUTRA_TestBody_Set ( true );

    //Set RS EPRE and parameters according to T1

    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT1);

    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );

    //@siclog "Step 2-9" siclog@

    f_EUTRA_ServiceRequestAndActivate_SRB2_DRB ( eutra_Cell2 );

    //@siclog "Step 9A-9B" siclog@

    f_CSFBParametersProc_1xRTT(eutra_Cell2, px_RAND2);

    //@siclog steps 10-end of preregistration siclog@

    f_CDMA2000_PreRegistration (eutra_Cell2);

    //Wait for the end of the test body

    f_IRAT_WaitForCoOrd_Trigger ( CDMA2000 );

    f_EUTRA_TestBody_Set ( false );

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    //Remove the non-serving cells

    f_EUTRA_Postamble ( eutra_Cell2, E2_CONNECTED );
  } // end of test case 13.4.4.1



After change :

	     function f_TC_13_4_4_1_EUTRA ()  runs on EUTRA_PTC

  { /* Pre-registration at 1xRTT and Cell reselection / 1x Zone Registration */

    var IRAT_CoOrd_SysInfo_Type v_CDMA2000SysInfo1;

    var IRAT_CoOrd_SysInfo_Type v_CDMA2000SysInfo2;

    var template (value) CellPowerList_Type v_CellPowerList_AtT1;

    v_CellPowerList_AtT1 := {

      cs_CellPower (eutra_Cell2, -79)

    };

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init( c6  );

    //Set maximum cell powel level for Cell 1

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell1, -60 );

    f_EUTRA_CellInfo_InitMaxReferencePower ( eutra_Cell2, -60 );

    // Wait for CDMA2000Sysinfo to fetch color code

    v_CDMA2000SysInfo1 := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );

    v_CDMA2000SysInfo2 := f_IRAT_WaitForCoOrd_SysInfo ( CDMA2000 );

    f_InitialiseSib8_1XRTT_PreRegistration (eutra_Cell1, v_CDMA2000SysInfo1, px_1XRTT_Zone_Timer);

    f_InitialiseSib8_1XRTT_PreRegistration (eutra_Cell2, v_CDMA2000SysInfo2, px_1XRTT_Zone_Timer);

    //Create and configure Cell 1

    f_EUTRA_CellConfig_Def ( eutra_Cell1 );

    f_EUTRA_CellConfig_Def ( eutra_Cell2 );

    f_EUTRA_SetCellPower( eutra_Cell1, -85 );

    //Preamble to ensure USIM is updated with valid TAI and to bring UE to initial state

    f_EUTRA_Preamble_CDMA2000_PreRegistration(eutra_Cell1, STATE2_IDLEUPDATE, true, px_RAND);

    //Send trigger to inform start of Test Body

    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );

    f_EUTRA_TestBody_Set ( true );

    //Set RS EPRE and parameters according to T1

    f_EUTRA_SetCellPowerList (v_CellPowerList_AtT1);

    f_IRAT_SendCoOrd ( CDMA2000, cms_IRAT_Trigger );

    //@siclog "Step 2-9" siclog@

    f_EUTRA_ServiceRequestAndActivate_SRB2_DRB ( eutra_Cell2 );

    //@siclog "Step 9A-9B" siclog@

    f_CSFBParametersProc_1xRTT(eutra_Cell2, px_RAND2);

    //@siclog steps 10-end of preregistration siclog@

    f_CDMA2000_PreRegistration (eutra_Cell2);

    //Wait for the end of the test body

    f_IRAT_WaitForCoOrd_Trigger ( CDMA2000 );

    f_EUTRA_TestBody_Set ( false );

    //test body ends

    ///////////////////////////////////////////////////////////////////////////////////

    //      Postamble

    //////////////////////////////////////////////////////////////////////////////////

    //Remove the non-serving cells


f_EUTRA_Postamble_TunneledDetach ( eutra_Cell2, E2_CONNECTED );
  } // end of test case 13.4.4.1



Change 2

	Function name
	f_EUTRA_Postamble_TunneledDetach

	Reason for change
	New function to handle tunnelled detach as proposed by MCC160 in comments to R5s140804.

Command for “SWITCH_OFF” needs to be called before handling of power down registration so that the procedures can be triggered.


	Summary of change
	New function to handle tunnelled detach.

	Source of change
	Common/EUTRA/EUTRA_CommonProcedures.ttcn


NEW :

	  function f_EUTRA_Postamble_TunneledDetach(EUTRA_CellId_Type p_CellId,

                             EUTRA_POSTAMBLE_STATE_Type p_E_State,

                             EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                             RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                             integer p_SelectedPlmnId := 1,








 CDMA2000_Type p_CDMA2000Type := type1XRTT) runs on EUTRA_PTC

  { /* @sic This function has been completely changed in R5-103847 sic@ */

    // @sic R5s11046: new parameter p_SelectedPlmnId sic@

   var CDMA2000_Type v_CDMA2000Type;

   if (pc_SwitchOnOff) // The UE can actually be switched off

      {


    f_UT_SwitchOffUE (UT, false); // @sic R5s100584 sic@ 

        select (p_E_State) {

          case (E1_IDLE)

          {

            // Not Handled yet




f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "");

          }

          case (E2_CONNECTED, E3_TESTMODE)

          {

            //Expect the UE to detach and release RRC connection




if (p_CDMA2000Type == type1XRTT)




{





f_CSFBParametersProc_1xRTT(p_CellId, px_RAND);

                f_CDMA2000_PowerDownDetach_1xRTT(p_CellId);




} else // HRPD




{





// Not Handled yet




     f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "");




}

            f_EUTRA_UE_TunneledDetach_SwitchOffUe(p_CellId, RRC_CONNECTED, p_RRC_TI, p_ForcedAttach, omit, p_SelectedPlmnId); // @sic R5s110176, R5s110468 sic@

          }

          case (E2_T3440) // @sic R5s110007 sic@

          {

            // Not Handled yet




f_EUTRA_SetVerdictFailOrInconc (__FILE__, __LINE__, "");

          }

          case (E4_DEREGISTERED)

          {

            f_UT_SwitchOffUE (UT, false);

          }

        }

      }

    else

      {// The power must be removed from the UE

        f_UT_PowerOffUE (UT);

      }

    //And now remove the cell(s)

    f_EUTRA_ReleaseAllCells();

  } // end of f_EUTRA_Postamble



Change 3

	Function name
	f_EUTRA_UE_TunneledDetach_SwitchOffUe

	Reason for change
	The existing function “f_EUTRA_UE_Detach_SwitchOffUe” has a “SWITCH_OFF” command which triggers the detach procedure. This is not required as the switch off command will now be performed in the new function “f_EUTRA_Postamble_TunneledDetach”. 
A new function based on f_EUTRA_UE_Detach_SwitchOffUe with switch off command removed is introduced.

	Summary of change
	New function based on f_EUTRA_UE_Detach_SwitchOffUe with switch off command removed.

	Source of change
	Common/EUTRA/EUTRA_NASSteps.ttcn


NEW :

	       //----------------------------------------------------------------------------

  /*

   * @desc      This function is used in tunnelled cases where detach takes place on CDMA2000 cell first, so switch off command is not required in this function. Waits for a Detach message and releases the RRC connection

   * @param     p_CellId

   * @param     p_RRCState

   * @param     p_RRC_TI            (default value: tsc_RRC_TI_Def)

   * @param     p_ForcedAttach      (default value: NORMAL)

   * @param     p_DetachType        (default value: omit - which means it will be determind by the function);  @sic R5-103675 sic@

   * @param     p_SelectedPlmnId    (default value: 1)    .. selectedPLMN-Identity as used in RRCConnectionSetupComplete (starting with 1)

   * @param     p_USIM_Rmvd

   * @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A, LTE_IRAT, POS)

   */

  function f_EUTRA_UE_TunneledDetach_SwitchOffUe(EUTRA_CellId_Type p_CellId,

                                         EUTRA_RRC_STATE_Type p_RRCState,

                                         RRC_TransactionIdentifier p_RRC_TI := tsc_RRC_TI_Def,

                                         EUTRA_ATTACH_Type p_ForcedAttach := NORMAL,

                                         template DetachType p_DetachType := omit,

                                         integer p_SelectedPlmnId := 1,

                                         boolean p_USIM_Rmvd := false) runs on EUTRA_PTC

  { // @sic R5s11046: new parameter p_SelectedPlmnId sic@

    var GutiParameters_Type v_Guti := f_EUTRA_CellInfo_GetGuti(p_CellId);

    var template (present) NAS_MSG_Indication_Type v_ExpectedDetachRequest, v_ExpectedServiceRequest;

    var NAS_MSG_Indication_Type v_NasInd;

    var template (present) DetachType v_DetachType;

    var NAS_AttDetValue_Type v_AttachType := f_GetEDetachType(p_ForcedAttach);  // @sic R5s100538, R5s110176 sic@

    var NAS_KsiValue v_KsiValue := f_EUTRA_SecurityKSIasme_Get();

    if (ispresent(p_DetachType.typeOfDetach)) {  // @sic R5-103675 sic@

      v_DetachType := p_DetachType;     // compiler warning cannot be suppressed as there is no way to convert 'template' to 'template (present)'

    } else {

      v_DetachType := cr_DetachType('1'B, v_AttachType);

    }

    v_ExpectedDetachRequest :=

      cr_NAS_Indication((tsc_SHT_IntegrityProtected, tsc_SHT_IntegrityProtected_Ciphered),

                          cr_508_DETACH_REQUEST_MO(cr_NAS_KeySetIdentifier (?, ?), // @sic R5s110031 sic@

                                                   f_GutiParameters2MobileIdentity(omit, v_Guti),  // @sic R5s090225, R5s090310 sic@

                                                   v_DetachType)); // @sic R5-103675 sic@

    v_ExpectedServiceRequest := cr_NAS_Indication(tsc_SHT_IntegrityProtected, cr_508_SERVICE_REQUEST(v_KsiValue));

      if (pc_SwitchOnOff or p_USIM_Rmvd and pc_USIM_Removal) // The UE can be switched off @sic R5s09225 sic@ or USIM removed @sic R5s120597r1 sic@

        {

          select (p_RRCState) {

            case (RRC_IDLE) {

              v_NasInd := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                                         p_RRC_TI, ?, // @sic R5s100766 sic@

                                                         (v_ExpectedDetachRequest, v_ExpectedServiceRequest), // @sic R5-131868 sic@

                                                         p_SelectedPlmnId); // @sic R5s110468 sic@

              if (ischosen(v_NasInd.Pdu.Msg.sERVICE_REQUEST)) { // @sic R5-131868 sic@

                f_EUTRA_Activate_SRB2_DRB_SendRrcMsg(p_CellId,

                                                     p_RRC_TI,

                                                     v_NasInd.SecurityProtection.NasCount,

                                                     0,

                                                     0,

                                                     omit,

                                                     omit,

                                                     omit,

                                                     omit,

                                                     omit,

                                                     true,

                                                     omit);

                f_EUTRA_Activate_SRB2_DRB_ReceiveRrcAndNasMsg(p_CellId,

                                                              p_RRC_TI,

                                                              0,

                                                              0);

                // DRB configured, so when the UE has finished Deregistration on IMS, it will then send the Detach..

                SRB.receive(car_SRB2_NasPdu_IND(p_CellId, v_ExpectedDetachRequest));

              }

            }

            case (RRC_CONNECTED) {

              alt {

                [] SRB.receive(car_SRB1_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

                [] SRB.receive(car_SRB2_NasPdu_IND(p_CellId, v_ExpectedDetachRequest)) { }

              }

            }

          }

          //Now the connection can be released

          f_EUTRA_RRC_ConnectionRelease(p_CellId, p_RRC_TI);

        }

      else // @sic R5s100584 sic@

        {// The power must be removed from the UE

          f_UT_PowerOffUE (UT);

          //Reset all RBs

          f_EUTRA_SS_SRBs_DRBs_Reset(p_CellId);   // @sic R5s110423 sic@

        }

  } //end of f_EUTRA_UE_TunneledDetach_SwitchOffUe

  


Change 4
	Function name
	f_TC_13_4_4_1_CDMA2000

	Reason for change
	Postamble modified to allow UE to perform Power Down Registration procedure when the UE supports being switched off (corresponding change for postamble modification in EUTRA_PTC in Change 1 above).

	Summary of change
	Added function call to “f_RTT1X_DetachOnSwitchOff_PreReg”, when “pc_SwitchOnOff” is TRUE, to handle system command and indication.

	Source of change
	LTE_IRAT\13\MultiLayer_Procedures_EC2K_CDMA2000.ttcn


Before change:
	  function f_TC_13_4_4_1_CDMA2000 () runs on CDMA2000_PTC

  { /* Pre-registration at 1xRTT and Cell reselection / 1x Zone Registration */

    ///Initialise test case

    f_CDMA2000_Init ( EUTRA_CDMA2000, type1XRTT );//@sic R5s120891 sic@

    //Send Sys info for ARFCN

    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell19 );

    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell21 );

    //Create CDMA2000 cell19 cell20 and broadcast System Infomation

    f_CDMA2000CreateCell ( cdma2000_Cell19 );

    f_CDMA2000CreateCell ( cdma2000_Cell21 );

    f_CDMA2000_PreRegister( cdma2000_Cell19,  px_RAND );

    //First Trigger: to start test case

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_CDMA2000_TestBody_Set( true );

    //------------------------------------------------

    //Start TestBody

    //@siclog "Step 3" siclog@

    f_CDMA2000_PreRegister (cdma2000_Cell21,  px_RAND2,zoneBasedAttach); //@sic R5-130660 R5s120891 R5s 140332 sic@

    //End TestBody

    //------------------------------------------------

    f_IRAT_SendCoOrd ( EUTRA, cms_IRAT_Trigger );

    f_CDMA2000_TestBody_Set ( false );

    //  release cells

    f_CDMA2000_ReleaseCells();

  }; //end of testcase 13.4.4.1


After change:

	    function f_TC_13_4_4_1_CDMA2000 () runs on CDMA2000_PTC

  { /* Pre-registration at 1xRTT and Cell reselection / 1x Zone Registration */

    ///Initialise test case

    f_CDMA2000_Init ( EUTRA_CDMA2000, type1XRTT );//@sic R5s120891 sic@

    //Send Sys info for ARFCN

    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell19 );

    f_CDMA2000_Send_IR_SysInfoToEUTRA (cdma2000_Cell21 );

    //Create CDMA2000 cell19 cell20 and broadcast System Infomation

    f_CDMA2000CreateCell ( cdma2000_Cell19 );

    f_CDMA2000CreateCell ( cdma2000_Cell21 );

    f_CDMA2000_PreRegister( cdma2000_Cell19,  px_RAND );

    //First Trigger: to start test case

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    f_CDMA2000_TestBody_Set( true );

    //------------------------------------------------

    //Start TestBody

    //@siclog "Step 3" siclog@

    f_CDMA2000_PreRegister (cdma2000_Cell21,  px_RAND2,zoneBasedAttach); //@sic R5-130660 R5s120891 R5s 140332 sic@

    //End TestBody

    //------------------------------------------------

    f_IRAT_SendCoOrd ( EUTRA, cms_IRAT_Trigger );

    f_CDMA2000_TestBody_Set ( false );


if (pc_SwitchOnOff) // The UE can actually be switched off

    {


  //Switch off the UE


f_RTT1X_DetachOnSwitchOff_PreReg(cdma2000_Cell19);


}
    //  release cells

    f_CDMA2000_ReleaseCells();

  }; //end of testcase 13.4.4.1

